Rules for integrands of the form $(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n$

1. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2+b^2==0$

1: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2+b^2==0\,\land\,m\in\mathbb{Z}$

- **Derivation: Algebraic simplification**
- **Basis: If** $b\,c+a\,d==0\,\land\,a^2+b^2==0$**, then** $(a+b\,\text{Tan}[z])\,(c+d\,\text{Tan}[z])==a\,c\,\text{Sec}[z]^2$
- **Rule: If** $b\,c+a\,d==0\,\land\,a^2+b^2==0\,\land\,m\in\mathbb{Z}$**, then**

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx\,\longrightarrow\,a^m\,c^m\int \text{Sec}[e+f\,x]^{2\,m}\,(c+d\,\text{Tan}[e+f\,x])^{n-m}\,dx$$

- **Program code:**

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_.*(c_+d_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  a^m*c^m*Int[Sec[e+f*x]^(2*m)*(c+d*Tan[e+f*x])^(n-m),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2+b^2,0] && IntegerQ[m] && Not[IGtQ[n,0] && (LtQ[m,0] || GtQ[m,n])]
```

2: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c+a\,d==0\,\land\,a^2+b^2==0$

- **Derivation: Integration by substitution**
- **Basis: If** $b\,c+a\,d==0\,\land\,a^2+b^2==0$**, then**
$(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n==\frac{a\,c}{f}\,\text{Subst}\left[(a+b\,x)^{m-1}\,(c+d\,x)^{n-1},\,x,\,\text{Tan}[e+f\,x]\right]\,\partial_x\text{Tan}[e+f\,x]$
- **Rule: If** $b\,c+a\,d==0\,\land\,a^2+b^2==0$**, then**

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx\,\longrightarrow\,\frac{a\,c}{f}\,\text{Subst}\left[\int (a+b\,x)^{m-1}\,(c+d\,x)^{n-1}\,dx,\,x,\,\text{Tan}[e+f\,x]\right]$$

- **Program code:**

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a*c/f*Subst[Int[(a+b*x)^(m-1)*(c+d*x)^(n-1),x],x,Tan[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2+b^2,0]
```

2. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0$

1. $\int (a+b\,\text{Tan}[e+f\,x])\,(c+d\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0$

1: $\int (a+b\,\text{Tan}[e+f\,x])\,(c+d\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,b\,c+a\,d==0$

- **Derivation: Tangent recurrence 2b with** $A\to a^2,\ B\to 2\,a\,b,\ C\to b^2,\ m\to -1,\ n\to 1$
- **Rule: If** $b\,c-a\,d\neq 0\,\wedge\,b\,c+a\,d==0$, **then**

$$\int (a+b\,\text{Tan}[e+f\,x])\,(c+d\,\text{Tan}[e+f\,x])\,dx\ \longrightarrow\ (a\,c-b\,d)\,x+\frac{b\,d\,\text{Tan}[e+f\,x]}{f}$$

- **Program code:**

```
Int[(a_+b_.*tan[e_.+f_.*x_])*(c_+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (a*c-b*d)*x + b*d*Tan[e+f*x]/f /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[b*c+a*d,0]
```

2: $\int (a+b\,\text{Tan}[e+f\,x])\,(c+d\,\text{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,b\,c+a\,d\neq 0$

- **Derivation: Tangent recurrence 2b with** $A\to a^2,\ B\to 2\,a\,b,\ C\to b^2,\ m\to -1,\ n\to 1$
- **Rule: If** $b\,c-a\,d\neq 0\,\wedge\,b\,c+a\,d\neq 0$, **then**

$$\int (a+b\,\text{Tan}[e+f\,x])\,(c+d\,\text{Tan}[e+f\,x])\,dx\ \longrightarrow\ (a\,c-b\,d)\,x+\frac{b\,d\,\text{Tan}[e+f\,x]}{f}+(b\,c+a\,d)\int \text{Tan}[e+f\,x]\,dx$$

- **Program code:**

```
Int[(a_+b_.*tan[e_.+f_.*x_])*(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (a*c-b*d)*x + b*d*Tan[e+f*x]/f + (b*c+a*d)*Int[Tan[e+f*x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[b*c+a*d,0]
```

2. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0$

1: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,m<0$

Derivation: Symmetric tangent recurrence 2a with $A\to c,\ B\to d,\ n\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,m<0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx\,\longrightarrow\,-\frac{(b\,c-a\,d)\,(a+b\,\mathrm{Tan}[e+f\,x])^m}{2\,a\,f\,m}+\frac{b\,c+a\,d}{2\,a\,b}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  -(b*c-a*d)*(a+b*Tan[e+f*x])^m/(2*a*f*m) +
  (b*c+a*d)/(2*a*b)*Int[(a+b*Tan[e+f*x])^(m+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && LtQ[m,0]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,m\nless 0$

Derivation: Symmetric tangent recurrence 3a with $A\to c,\ B\to d,\ n\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,m\nless 0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx\,\longrightarrow\,\frac{d\,(a+b\,\mathrm{Tan}[e+f\,x])^m}{f\,m}+\frac{b\,c+a\,d}{b}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  d*(a+b*Tan[e+f*x])^m/(f*m) + (b*c+a*d)/b*Int[(a+b*Tan[e+f*x])^m,x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && Not[LtQ[m,0]]
```

3. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0$

1: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,m>0$

Derivation: Tangent recurrence 2a with $A\to 0,\ B\to A,\ C\to B,\ n\to -1$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,m>0$, then

$$\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x]) \, dx \longrightarrow$$
$$\frac{d (a + b \operatorname{Tan}[e + f x])^m}{f m} + \int (a + b \operatorname{Tan}[e + f x])^{m-1} (a c - b d + (b c + a d) \operatorname{Tan}[e + f x]) \, dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  d*(a+b*Tan[e+f*x])^m/(f*m) +
  Int[(a+b*Tan[e+f*x])^(m-1)*Simp[a*c-b*d+(b*c+a*d)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && GtQ[m,0]
```

2: $\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x]) \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land m < -1$

Derivation: Tangent recurrence 1b with $A \to c, B \to d, C \to 0, n \to 0$

Rule: If $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land m < -1$**, then**

$$\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x]) \, dx \longrightarrow$$
$$\frac{(b c - a d) (a + b \operatorname{Tan}[e + f x])^{m+1}}{f (m + 1) \left(a^2 + b^2\right)} + \frac{1}{a^2 + b^2} \int (a + b \operatorname{Tan}[e + f x])^{m+1} (a c + b d - (b c - a d) \operatorname{Tan}[e + f x]) \, dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (b*c-a*d)*(a+b*Tan[e+f*x])^(m+1)/(f*(m+1)*(a^2+b^2)) +
  1/(a^2+b^2)*Int[(a+b*Tan[e+f*x])^(m+1)*Simp[a*c+b*d-(b*c-a*d)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && LtQ[m,-1]
```

3. $\int \frac{c + d \operatorname{Tan}[e + f x]}{a + b \operatorname{Tan}[e + f x]} \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0$

1: $\int \frac{c + d \operatorname{Tan}[e + f x]}{a + b \operatorname{Tan}[e + f x]} \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land a c + b d == 0$

Derivation: Algebraic expansion and reciprocal for integration

Basis: If $a c + b d == 0$**, then** $\frac{c+d \operatorname{Tan}[z]}{a+b \operatorname{Tan}[z]} == \frac{c (b \operatorname{Cos}[z]-a \operatorname{Sin}[z])}{b (a \operatorname{Cos}[z]+b \operatorname{Sin}[z])}$

Rule: If $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land a c + b d == 0$**, then**

$$\int \frac{c + d\,\text{Tan}[e + f\,x]}{a + b\,\text{Tan}[e + f\,x]}\,dx \longrightarrow \frac{c}{b}\int \frac{b\,\text{Cos}[e + f\,x] - a\,\text{Sin}[e + f\,x]}{a\,\text{Cos}[e + f\,x] + b\,\text{Sin}[e + f\,x]}\,dx \longrightarrow \frac{c}{b\,f}\,\text{Log}[a\,\text{Cos}[e + f\,x] + b\,\text{Sin}[e + f\,x]]$$

Program code:

```
Int[(c_+d_.*tan[e_.+f_.*x_])/(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  c/(b*f)*Log[RemoveContent[a*Cos[e+f*x]+b*Sin[e+f*x],x]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && EqQ[a*c+b*d,0]
```

2: $\int \frac{c + d\,\text{Tan}[e + f\,x]}{a + b\,\text{Tan}[e + f\,x]}\,dx$ when $b\,c - a\,d \neq 0 \wedge a^2 + b^2 \neq 0 \wedge a\,c + b\,d \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{c+d\,z}{a+b\,z} == \frac{a\,c+b\,d}{a^2+b^2} + \frac{(b\,c-a\,d)\,(b-a\,z)}{(a^2+b^2)\,(a+b\,z)}$

Rule: If $b\,c - a\,d \neq 0 \wedge a^2 + b^2 \neq 0 \wedge a\,c + b\,d \neq 0$**, then**

$$\int \frac{c + d\,\text{Tan}[e + f\,x]}{a + b\,\text{Tan}[e + f\,x]}\,dx \longrightarrow \frac{(a\,c + b\,d)\,x}{a^2 + b^2} + \frac{b\,c - a\,d}{a^2 + b^2}\int \frac{b - a\,\text{Tan}[e + f\,x]}{a + b\,\text{Tan}[e + f\,x]}\,dx$$

Program code:

```
Int[(c_.+d_.*tan[e_.+f_.*x_])/(a_.+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (a*c+b*d)*x/(a^2+b^2) + (b*c-a*d)/(a^2+b^2)*Int[(b-a*Tan[e+f*x])/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[a*c+b*d,0]
```

4. $\int \frac{c + d\,\text{Tan}[e + f\,x]}{\sqrt{a + b\,\text{Tan}[e + f\,x]}}\,dx$ when $b\,c - a\,d \neq 0 \wedge a^2 + b^2 \neq 0$

1. $\int \frac{c + d\,\text{Tan}[e + f\,x]}{\sqrt{b\,\text{Tan}[e + f\,x]}}\,dx$

1: $\int \frac{c + d\,\text{Tan}[e + f\,x]}{\sqrt{b\,\text{Tan}[e + f\,x]}}\,dx$ when $c^2 - d^2 == 0$

Derivation: Integration by substitution

Basis: If $c^2 - d^2 == 0$**, then** $\frac{c+d\,\text{Tan}[e+f\,x]}{\sqrt{b\,\text{Tan}[e+f\,x]}} == -\frac{2\,c^2}{f}\,\text{Subst}\left[\frac{1}{2\,c\,d+b\,x^2},\ x,\ \frac{c-d\,\text{Tan}[e+f\,x]}{\sqrt{b\,\text{Tan}[e+f\,x]}}\right]\partial_x \frac{c-d\,\text{Tan}[e+f\,x]}{\sqrt{b\,\text{Tan}[e+f\,x]}}$

Rule: If $c^2 - d^2 == 0$**, then**

$$\int \frac{c+d\,\mathrm{Tan}[e+f\,x]}{\sqrt{b\,\mathrm{Tan}[e+f\,x]}}\,dx \to -\frac{2\,d^2}{f}\,\mathrm{Subst}\Big[\int \frac{1}{2\,c\,d+b\,x^2}\,dx,\,x,\,\frac{c-d\,\mathrm{Tan}[e+f\,x]}{\sqrt{b\,\mathrm{Tan}[e+f\,x]}}\Big]$$

Program code:

```
Int[(c_+d_.*tan[e_.+f_.*x_])/Sqrt[b_.*tan[e_.+f_.*x_]],x_Symbol] :=
  -2*d^2/f*Subst[Int[1/(2*c*d+b*x^2),x],x,(c-d*Tan[e+f*x])/Sqrt[b*Tan[e+f*x]]] /;
FreeQ[{b,c,d,e,f},x] && EqQ[c^2-d^2,0]
```

2. $\int \frac{c+d\,\mathrm{Tan}[e+f\,x]}{\sqrt{b\,\mathrm{Tan}[e+f\,x]}}\,dx$ when $c^2-d^2\neq 0$

x: $\int \frac{c+d\,\mathrm{Tan}[e+f\,x]}{\sqrt{b\,\mathrm{Tan}[e+f\,x]}}\,dx$ when $c^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $c+d\,z == \frac{(c+d)\,(1+z)}{2}+\frac{(c-d)\,(1-z)}{2}$

Rule: If $c^2-d^2\neq 0$, **then**

$$\int \frac{c+d\,\mathrm{Tan}[e+f\,x]}{\sqrt{b\,\mathrm{Tan}[e+f\,x]}}\,dx \to \frac{c+d}{2}\int \frac{1+\mathrm{Tan}[e+f\,x]}{\sqrt{b\,\mathrm{Tan}[e+f\,x]}}\,dx+\frac{c-d}{2}\int \frac{1-\mathrm{Tan}[e+f\,x]}{\sqrt{b\,\mathrm{Tan}[e+f\,x]}}\,dx$$

Program code:

```
(* Int[(c_+d_.*tan[e_.+f_.*x_])/Sqrt[b_.*tan[e_.+f_.*x_]],x_Symbol] :=
  (c+d)/2*Int[(1+Tan[e+f*x])/Sqrt[b*Tan[e+f*x]],x] +
  (c-d)/2*Int[(1-Tan[e+f*x])/Sqrt[b*Tan[e+f*x]],x] /;
FreeQ[{b,c,d,e,f},x] && NeQ[c^2+d^2,0] && NeQ[c^2-d^2,0] *)
```

1: $\int \frac{c+d\,\mathrm{Tan}[e+f\,x]}{\sqrt{b\,\mathrm{Tan}[e+f\,x]}}\,dx$ when $c^2+d^2==0$

Derivation: Integration by substitution

Basis: If $c^2+d^2==0$, **then** $\frac{c+d\,\mathrm{Tan}[e+f\,x]}{\sqrt{b\,\mathrm{Tan}[e+f\,x]}} == \frac{2\,c^2}{f}\,\mathrm{Subst}\Big[\frac{1}{b\,c-d\,x^2},\,x,\,\sqrt{b\,\mathrm{Tan}[e+f\,x]}\Big]\,\partial_x\sqrt{b\,\mathrm{Tan}[e+f\,x]}$

Note: This is just a special case of the following rule, but it saves two steps by canceling out the gcd.

Rule: If $c^2+d^2==0$, **then**

$$\int \frac{c+d\,\text{Tan}[e+f\,x]}{\sqrt{b\,\text{Tan}[e+f\,x]}}\,dx \longrightarrow \frac{2\,c^2}{f}\,\text{Subst}\Big[\int \frac{1}{b\,c-d\,x^2}\,dx,\ x,\ \sqrt{b\,\text{Tan}[e+f\,x]}\Big]$$

Program code:

```
Int[(c_+d_.*tan[e_.+f_.*x_])/Sqrt[b_.*tan[e_.+f_.*x_]],x_Symbol] :=
  2*c^2/f*Subst[Int[1/(b*c-d*x^2),x],x,Sqrt[b*Tan[e+f*x]]] /;
FreeQ[{b,c,d,e,f},x] && EqQ[c^2+d^2,0]
```

x: $\int \frac{c+d\,\text{Tan}[e+f\,x]}{\sqrt{b\,\text{Tan}[e+f\,x]}}\,dx$ when $c^2-d^2\neq 0\land c^2+d^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $c+d\,z == \frac{(c+i\,d)}{2}\,(1-i\,z)+\frac{(c-i\,d)}{2}\,(1+i\,z)$

Note: Introduces the imaginary unit.

- **Rule: If** $c^2-d^2\neq 0\land c^2+d^2\neq 0$**, then**

$$\int \frac{c+d\,\text{Tan}[e+f\,x]}{\sqrt{b\,\text{Tan}[e+f\,x]}}\,dx \longrightarrow \frac{(c+i\,d)}{2}\int \frac{1-i\,\text{Tan}[e+f\,x]}{\sqrt{b\,\text{Tan}[e+f\,x]}}\,dx+\frac{(c-i\,d)}{2}\int \frac{1+i\,\text{Tan}[e+f\,x]}{\sqrt{b\,\text{Tan}[e+f\,x]}}\,dx$$

Program code:

```
(* Int[(c_+d_.*tan[e_.+f_.*x_])/Sqrt[b_.*tan[e_.+f_.*x_]],x_Symbol] :=
  (c+I*d)/2*Int[(1-I*Tan[e+f*x])/Sqrt[b*Tan[e+f*x]],x] + (c-I*d)/2*Int[(1+I*Tan[e+f*x])/Sqrt[b*Tan[e+f*x]],x] /;
FreeQ[{b,c,d,e,f},x] && NeQ[c^2-d^2,0] && NeQ[c^2+d^2,0] *)
```

2: $\int \frac{c + d\,\text{Tan}[e + f x]}{\sqrt{b\,\text{Tan}[e + f x]}}\,dx$ when $c^2 - d^2 \neq 0 \land c^2 + d^2 \neq 0$

Derivation: Integration by substitution

- **Basis:** $\frac{c+d\,\text{Tan}[e+f x]}{\sqrt{b\,\text{Tan}[e+f x]}} == \frac{2}{f}\,\text{Subst}\left[\frac{b c+d x^2}{b^2+x^4},\, x,\, \sqrt{b\,\text{Tan}[e+f x]}\right] \partial_x \sqrt{b\,\text{Tan}[e+f x]}$

- **Rule: If** $c^2 - d^2 \neq 0 \land c^2 + d^2 \neq 0$, **then**

$$\int \frac{c + d\,\text{Tan}[e + f x]}{\sqrt{b\,\text{Tan}[e + f x]}}\,dx \longrightarrow \frac{2}{f}\,\text{Subst}\left[\int \frac{b c + d x^2}{b^2 + x^4}\,dx,\, x,\, \sqrt{b\,\text{Tan}[e + f x]}\right]$$

Program code:

```
Int[(c_+d_.*tan[e_.+f_.*x_])/Sqrt[b_.*tan[e_.+f_.*x_]],x_Symbol] :=
  2/f*Subst[Int[(b*c+d*x^2)/(b^2+x^4),x],x,Sqrt[b*Tan[e+f*x]]] /;
FreeQ[{b,c,d,e,f},x] && NeQ[c^2-d^2,0] && NeQ[c^2+d^2,0]
```

2. $\int \frac{c + d\,\text{Tan}[e + f x]}{\sqrt{a + b\,\text{Tan}[e + f x]}}\,dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0$

1: $\int \frac{c + d\,\text{Tan}[e + f x]}{\sqrt{a + b\,\text{Tan}[e + f x]}}\,dx$ when $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0 \land 2 a c d - b\left(c^2 - d^2\right) == 0$

Derivation: Integration by substitution

- **Basis: If** $2 a c d - b\left(c^2 - d^2\right) == 0$, **then** $\frac{c+d\,\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Tan}[e+f x]}} == -\frac{2 d^2}{f}\,\text{Subst}\left[\frac{1}{2 b c d-4 a d^2+x^2},\, x,\, \frac{b c-2 a d-b d\,\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Tan}[e+f x]}}\right] \partial_x \frac{b c-2 a d-b d\,\text{Tan}[e+f x]}{\sqrt{a+b\,\text{Tan}[e+f x]}}$

- **Rule: If** $b c - a d \neq 0 \land a^2 + b^2 \neq 0 \land c^2 + d^2 \neq 0 \land 2 a c d - b\left(c^2 - d^2\right) == 0$, **then**

$$\int \frac{c + d\,\text{Tan}[e + f x]}{\sqrt{a + b\,\text{Tan}[e + f x]}}\,dx \longrightarrow -\frac{2 d^2}{f}\,\text{Subst}\left[\int \frac{1}{2 b c d - 4 a d^2 + x^2}\,dx,\, x,\, \frac{b c - 2 a d - b d\,\text{Tan}[e + f x]}{\sqrt{a + b\,\text{Tan}[e + f x]}}\right]$$

Program code:

```
Int[(c_.+d_.*tan[e_.+f_.*x_])/Sqrt[a_+b_.*tan[e_.+f_.*x_]],x_Symbol] :=
  -2*d^2/f*Subst[Int[1/(2*b*c*d-4*a*d^2+x^2),x],x,(b*c-2*a*d-b*d*Tan[e+f*x])/Sqrt[a+b*Tan[e+f*x]]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && EqQ[2*a*c*d-b*(c^2-d^2),0]
```

2: $\int \frac{c+d\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\wedge a^2+b^2\neq 0\wedge c^2+d^2\neq 0\wedge 2\,a\,c\,d-b\,(c^2-d^2)\neq 0$

Derivation: Algebraic expansion

Note: The resulting integrands are of the form required by the above rule.

Rule: If $b\,c-a\,d\neq 0\wedge a^2+b^2\neq 0\wedge c^2+d^2\neq 0\wedge 2\,a\,c\,d-b\,(c^2-d^2)\neq 0$**, let** $q=\sqrt{a^2+b^2}$**, then**

$$\int \frac{c+d\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}\,dx\;\longrightarrow$$

$$\frac{1}{2\,q}\int \frac{a\,c+b\,d+c\,q+(b\,c-a\,d+d\,q)\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}\,dx-\frac{1}{2\,q}\int \frac{a\,c+b\,d-c\,q+(b\,c-a\,d-d\,q)\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}\,dx$$

Program code:

```
Int[(c_.+d_.*tan[e_.+f_.*x_])/Sqrt[a_+b_.*tan[e_.+f_.*x_]],x_Symbol] :=
  With[{q=Rt[a^2+b^2,2]},
  1/(2*q)*Int[(a*c+b*d+c*q+(b*c-a*d+d*q)*Tan[e+f*x])/Sqrt[a+b*Tan[e+f*x]],x] -
   1/(2*q)*Int[(a*c+b*d-c*q+(b*c-a*d-d*q)*Tan[e+f*x])/Sqrt[a+b*Tan[e+f*x]],x]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && NeQ[2*a*c*d-b*(c^2-d^2),0] &&
  (PerfectSquareQ[a^2+b^2] || RationalQ[a,b,c,d])
```

5: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\wedge a^2+b^2\neq 0\wedge c^2+d^2==0$

Derivation: Integration by substitution

Basis: If $c^2+d^2==0$**, then** $(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])==\frac{c\,d}{f}\,\mathrm{Subst}\left[\frac{\left(a+\frac{b\,x}{d}\right)^m}{d^2+c\,x},\,x,\,d\,\mathrm{Tan}[e+f\,x]\right]\,\partial_x\,(d\,\mathrm{Tan}[e+f\,x])$

Rule: If $b\,c-a\,d\neq 0\wedge a^2+b^2\neq 0\wedge c^2+d^2==0$**, then**

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx\;\longrightarrow\;\frac{c\,d}{f}\,\mathrm{Subst}\left[\int \frac{\left(a+\frac{b\,x}{d}\right)^m}{d^2+c\,x}\,dx,\,x,\,d\,\mathrm{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  c*d/f*Subst[Int[(a+b/d*x)^m/(d^2+c*x),x],x,d*Tan[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[c^2+d^2,0]
```

6. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0$

1: $\int (b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx$ when $c^2+d^2\neq 0\,\wedge\,2\,m\notin\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $(b\,z)^m\,(c+d\,z) == c\,(b\,z)^m+\frac{d}{b}\,(b\,z)^{m+1}$

Rule: If $c^2+d^2\neq 0\,\wedge\,2\,m\notin\mathbb{Z}$**, then**

$$\int (b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx \longrightarrow c\int (b\,\mathrm{Tan}[e+f\,x])^m\,dx+\frac{d}{b}\int (b\,\mathrm{Tan}[e+f\,x])^{m+1}\,dx$$

Program code:

```
Int[(b_.*tan[e_.+f_.*x_])^m_*(c_+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  c*Int[(b*Tan[e+f*x])^m,x] + d/b*Int[(b*Tan[e+f*x])^(m+1),x] /;
FreeQ[{b,c,d,e,f,m},x] && NeQ[c^2+d^2,0] && Not[IntegerQ[2*m]]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m\notin\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $c+d\,z == \frac{(c+i\,d)}{2}\,(1-i\,z)+\frac{(c-i\,d)}{2}\,(1+i\,z)$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m\notin\mathbb{Z}$**, then**

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])\,dx \longrightarrow$$
$$\frac{(c+i\,d)}{2}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(1-i\,\mathrm{Tan}[e+f\,x])\,dx+\frac{(c-i\,d)}{2}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(1+i\,\mathrm{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (c+I*d)/2*Int[(a+b*Tan[e+f*x])^m*(1-I*Tan[e+f*x]),x] +
  (c-I*d)/2*Int[(a+b*Tan[e+f*x])^m*(1+I*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && Not[IntegerQ[m]]
```

3. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0$

1. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m\leq -1$

1: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m\leq -1\,\wedge\,a^2+b^2==0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,m\leq -1\,\wedge\,a^2+b^2==0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^2\,dx\;\longrightarrow$$
$$-\frac{b\,(a\,c+b\,d)^2\,(a+b\,\mathrm{Tan}[e+f\,x])^m}{2\,a^3\,f\,m}+\frac{1}{2\,a^2}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\left(a\,c^2-2\,b\,c\,d+a\,d^2-2\,b\,d^2\,\mathrm{Tan}[e+f\,x]\right)dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^2,x_Symbol] :=
  -b*(a*c+b*d)^2*(a+b*Tan[e+f*x])^m/(2*a^3*f*m) +
  1/(2*a^2)*Int[(a+b*Tan[e+f*x])^(m+1)*Simp[a*c^2-2*b*c*d+a*d^2-2*b*d^2*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && LeQ[m,-1] && EqQ[a^2+b^2,0]
```

2. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m\leq -1\,\wedge\,a^2+b^2\neq 0$

1: $\int \frac{(c+d\,\mathrm{Tan}[e+f\,x])^2}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{(c+d\,z)^2}{a+b\,z}==\frac{d\,(2\,b\,c-a\,d)}{b^2}+\frac{d^2\,z}{b}+\frac{(b\,c-a\,d)^2}{b^2\,(a+b\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0$, then

$$\int \frac{(c+d\,\mathrm{Tan}[e+f\,x])^2}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx\;\longrightarrow\;\frac{d\,(2\,b\,c-a\,d)\,x}{b^2}+\frac{d^2}{b}\int \mathrm{Tan}[e+f\,x]\,dx+\frac{(b\,c-a\,d)^2}{b^2}\int\frac{1}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[(c_.+d_.*tan[e_.+f_.*x_])^2/(a_.+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  d*(2*b*c-a*d)*x/b^2 + d^2/b*Int[Tan[e+f*x],x] + (b*c-a*d)^2/b^2*Int[1/(a+b*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m<-1\,\wedge\,a^2+b^2\neq 0$

Derivation: Tangent recurrence 1b with $A\to c^2,\ B\to 2\,c\,d,\ C\to d^2,\ n\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,m<-1\,\wedge\,a^2+b^2\neq 0$**, then**

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^2\,dx\ \longrightarrow$$

$$\frac{(b\,c-a\,d)^2\,(a+b\,\mathrm{Tan}[e+f\,x])^{m+1}}{b\,f\,(m+1)\,\left(a^2+b^2\right)}+\frac{1}{a^2+b^2}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,\left(a\,c^2+2\,b\,c\,d-a\,d^2-\left(b\,c^2-2\,a\,c\,d-b\,d^2\right)\mathrm{Tan}[e+f\,x]\right)dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^2,x_Symbol] :=
  (b*c-a*d)^2*(a+b*Tan[e+f*x])^(m+1)/(b*f*(m+1)*(a^2+b^2)) +
  1/(a^2+b^2)*Int[(a+b*Tan[e+f*x])^(m+1)*Simp[a*c^2+2*b*c*d-a*d^2-(b*c^2-2*a*c*d-b*d^2)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && LtQ[m,-1] && NeQ[a^2+b^2,0]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^2\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m\nleq -1$

Derivation: Tangent recurrence 2b with $A\to c^2,\ B\to 2\,c\,d,\ C\to d^2,\ n\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,m\nleq -1$**, then**

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^2\,dx\ \longrightarrow\ \frac{d^2\,(a+b\,\mathrm{Tan}[e+f\,x])^{m+1}}{b\,f\,(m+1)}+\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,\left(c^2-d^2+2\,c\,d\,\mathrm{Tan}[e+f\,x]\right)dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^2,x_Symbol] :=
  d^2*(a+b*Tan[e+f*x])^(m+1)/(b*f*(m+1)) +
  Int[(a+b*Tan[e+f*x])^m*Simp[c^2-d^2+2*c*d*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && Not[LeQ[m,-1]] && Not[EqQ[m,2] && EqQ[a,0]]
```

4. $\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^n \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 == 0 \land c^2 + d^2 \neq 0$

   1. $\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^n \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 == 0 \land c^2 + d^2 \neq 0 \land m + n == 0$

      1. $\int (a + b \operatorname{Tan}[e + f x])^m (c + d \operatorname{Tan}[e + f x])^n \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 == 0 \land c^2 + d^2 \neq 0 \land m + n == 0 \land m \geq \frac{1}{2}$

         1: $\int \frac{\sqrt{a + b \operatorname{Tan}[e + f x]}}{\sqrt{c + d \operatorname{Tan}[e + f x]}} \, dx$ when $b c - a d \neq 0 \land a^2 + b^2 == 0 \land c^2 + d^2 \neq 0$

Derivation: Integration by substitution

- **Basis: If** $a^2 + b^2 == 0$, **then** $\frac{\sqrt{a+b \operatorname{Tan}[e+f x]}}{\sqrt{c+d \operatorname{Tan}[e+f x]}} == -\frac{2 a b}{f} \operatorname{Subst}\left[\frac{1}{a c-b d-2 a^2 x^2}, x, \frac{\sqrt{c+d \operatorname{Tan}[e+f x]}}{\sqrt{a+b \operatorname{Tan}[e+f x]}}\right] \partial_x \frac{\sqrt{c+d \operatorname{Tan}[e+f x]}}{\sqrt{a+b \operatorname{Tan}[e+f x]}}$

- **Rule: If** $b c - a d \neq 0 \land a^2 + b^2 == 0 \land c^2 + d^2 \neq 0$, **then**

$$\int \frac{\sqrt{a + b \operatorname{Tan}[e + f x]}}{\sqrt{c + d \operatorname{Tan}[e + f x]}} \, dx \longrightarrow -\frac{2 a b}{f} \operatorname{Subst}\left[\int \frac{1}{a c - b d - 2 a^2 x^2} \, dx, x, \frac{\sqrt{c + d \operatorname{Tan}[e + f x]}}{\sqrt{a + b \operatorname{Tan}[e + f x]}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*tan[e_.+f_.*x_]]/Sqrt[c_.+d_.*tan[e_.+f_.*x_]],x_Symbol] :=
  -2*a*b/f*Subst[Int[1/(a*c-b*d-2*a^2*x^2),x],x,Sqrt[c+d*Tan[e+f*x]]/Sqrt[a+b*Tan[e+f*x]]] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\bigwedge\,a^2+b^2==0\,\bigwedge\,c^2+d^2\neq 0\,\bigwedge\,m+n==0\,\bigwedge\,m>\frac{1}{2}$

Derivation: Symmetric tangent recurrence 1a with $A\to 1,\ B\to 0,\ n\to -m$

- **Note:** If $a^2+b^2==0\,\wedge\,c^2+d^2\neq 0$, then $a\,c-b\,d\neq 0$.

- **Rule:** If $b\,c-a\,d\neq 0\,\bigwedge\,a^2+b^2==0\,\bigwedge\,c^2+d^2\neq 0\,\bigwedge\,m+n==0\,\bigwedge\,m>\frac{1}{2}$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$
$$\frac{a\,b\,(a+b\,\mathrm{Tan}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{f\,(m-1)\,(a\,c-b\,d)}+\frac{2\,a^2}{a\,c-b\,d}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a*b*(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^(n+1)/(f*(m-1)*(a*c-b*d)) +
  2*a^2/(a*c-b*d)*Int[(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && EqQ[m+n,0] && GtQ[m,1/2]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\bigwedge\,a^2+b^2==0\,\bigwedge\,c^2+d^2\neq 0\,\bigwedge\,m+n==0\,\bigwedge\,m\leq -\frac{1}{2}$

Derivation: Symmetric tangent recurrence 2b with $A\to c,\ B\to d,\ n\to -m-1$

- **Note:** If $a^2+b^2==0\,\wedge\,c^2+d^2\neq 0$, then $a\,c-b\,d\neq 0$.

- **Rule:** If $b\,c-a\,d\neq 0\,\bigwedge\,a^2+b^2==0\,\bigwedge\,c^2+d^2\neq 0\,\bigwedge\,m+n==0\,\bigwedge\,m\leq -\frac{1}{2}$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$
$$\frac{a\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n}{2\,b\,f\,m}-\frac{a\,c-b\,d}{2\,b^2}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n-1}\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^n/(2*b*f*m) -
  (a*c-b*d)/(2*b^2)*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && EqQ[m+n,0] && LeQ[m,-1/2]
```

2. $\int (a + b\,\mathrm{Tan}[e + f\,x])^m\,(c + d\,\mathrm{Tan}[e + f\,x])^n\,dx$ when $b\,c - a\,d \neq 0\,\land\,a^2 + b^2 == 0\,\land\,c^2 + d^2 \neq 0\,\land\,m + n + 1 == 0$

1: $\int (a + b\,\mathrm{Tan}[e + f\,x])^m\,(c + d\,\mathrm{Tan}[e + f\,x])^n\,dx$ when $b\,c - a\,d \neq 0\,\land\,a^2 + b^2 == 0\,\land\,c^2 + d^2 \neq 0\,\land\,m + n + 1 == 0\,\land\,m < -1$

Derivation: Symmetric tangent recurrence 2b with $A \to 1,\ B \to 0,\ n \to -m - 1$

Rule: If $b\,c - a\,d \neq 0\,\land\,a^2 + b^2 == 0\,\land\,c^2 + d^2 \neq 0\,\land\,m + n + 1 == 0\,\land\,m < -1$, then

$$\int (a + b\,\mathrm{Tan}[e + f\,x])^m\,(c + d\,\mathrm{Tan}[e + f\,x])^n\,dx \longrightarrow$$

$$\frac{a\,(a + b\,\mathrm{Tan}[e + f\,x])^m\,(c + d\,\mathrm{Tan}[e + f\,x])^{n+1}}{2\,f\,m\,(b\,c - a\,d)} + \frac{1}{2\,a}\int (a + b\,\mathrm{Tan}[e + f\,x])^{m+1}\,(c + d\,\mathrm{Tan}[e + f\,x])^n\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(2*f*m*(b*c-a*d)) +
  1/(2*a)*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n,x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && EqQ[m+n+1,0] && LtQ[m,-1]
```

2: $\int (a + b\,\mathrm{Tan}[e + f\,x])^m\,(c + d\,\mathrm{Tan}[e + f\,x])^n\,dx$ when $b\,c - a\,d \neq 0\,\land\,a^2 + b^2 == 0\,\land\,c^2 + d^2 \neq 0\,\land\,m + n + 1 == 0\,\land\,m \nless -1$

Derivation: Symmetric tangent recurrence 3b with $A \to 1,\ B \to 0,\ n \to -m - 1$

Rule: If $b\,c - a\,d \neq 0\,\land\,a^2 + b^2 == 0\,\land\,c^2 + d^2 \neq 0\,\land\,m + n + 1 == 0\,\land\,m \nless -1$, then

$$\int (a + b\,\mathrm{Tan}[e + f\,x])^m\,(c + d\,\mathrm{Tan}[e + f\,x])^n\,dx \longrightarrow$$

$$-\frac{d\,(a + b\,\mathrm{Tan}[e + f\,x])^m\,(c + d\,\mathrm{Tan}[e + f\,x])^{n+1}}{f\,m\,\left(c^2 + d^2\right)} + \frac{a}{a\,c - b\,d}\int (a + b\,\mathrm{Tan}[e + f\,x])^m\,(c + d\,\mathrm{Tan}[e + f\,x])^{n+1}\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  -d*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(f*m*(c^2+d^2)) +
  a/(a*c-b*d)*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && EqQ[m+n+1,0] && Not[LtQ[m,-1]]
```

3. $\int \frac{(c+d\,\mathrm{Tan}[e+f\,x])^n}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0$

1. $\int \frac{(c+d\,\mathrm{Tan}[e+f\,x])^n}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,n>0$

1: $\int \frac{(c+d\,\mathrm{Tan}[e+f\,x])^n}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,0<n<1$

Derivation: Symmetric tangent recurrence 2a with $A\to 1,\ B\to 0,\ m\to -1$

Derivation: Symmetric tangent recurrence 2b with $A\to c,\ B\to d,\ m\to -1,\ n\to n-1$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,0<n<1$**, then**

$$\int \frac{(c+d\,\mathrm{Tan}[e+f\,x])^n}{a+b\,\mathrm{Tan}[e+f\,x]}\,dx\,\longrightarrow$$
$$-\frac{(a\,c+b\,d)\,(c+d\,\mathrm{Tan}[e+f\,x])^n}{2\,(b\,c-a\,d)\,f\,(a+b\,\mathrm{Tan}[e+f\,x])}+$$
$$\frac{1}{2\,a\,(b\,c-a\,d)}\int (c+d\,\mathrm{Tan}[e+f\,x])^{n-1}\left(a\,c\,d\,(n-1)+b\,c^2+b\,d^2\,n-d\,(b\,c-a\,d)\,(n-1)\,\mathrm{Tan}[e+f\,x]\right)dx$$

Program code:

```
Int[(c_.+d_.*tan[e_.+f_.*x_])^n_/(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  -(a*c+b*d)*(c+d*Tan[e+f*x])^n/(2*(b*c-a*d)*f*(a+b*Tan[e+f*x])) +
  1/(2*a*(b*c-a*d))*Int[(c+d*Tan[e+f*x])^(n-1)*Simp[a*c*d*(n-1)+b*c^2+b*d^2*n-d*(b*c-a*d)*(n-1)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[0,n,1]
```

2: $\int \frac{(c+d\,\text{Tan}[e+f\,x])^n}{a+b\,\text{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq0\;\wedge\;a^2+b^2==0\;\wedge\;c^2+d^2\neq0\;\wedge\;n>1$

Derivation: Symmetric tangent recurrence 2a with $A\to c,\;B\to d,\;m\to-1,\;n\to n-1$

Rule: If $b\,c-a\,d\neq0\;\wedge\;a^2+b^2==0\;\wedge\;c^2+d^2\neq0\;\wedge\;n>1$**, then**

$$\int \frac{(c+d\,\text{Tan}[e+f\,x])^n}{a+b\,\text{Tan}[e+f\,x]}\,dx\;\longrightarrow$$
$$\frac{(b\,c-a\,d)\,(c+d\,\text{Tan}[e+f\,x])^{n-1}}{2\,a\,f\,(a+b\,\text{Tan}[e+f\,x])}+$$
$$\frac{1}{2\,a^2}\int(c+d\,\text{Tan}[e+f\,x])^{n-2}\left(a\,c^2+a\,d^2\,(n-1)-b\,c\,d\,n-d\,(a\,c\,(n-2)+b\,d\,n)\,\text{Tan}[e+f\,x]\right)dx$$

Program code:

```
Int[(c_.+d_.*tan[e_.+f_.*x_])^n_/(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  (b*c-a*d)*(c+d*Tan[e+f*x])^(n-1)/(2*a*f*(a+b*Tan[e+f*x])) +
  1/(2*a^2)*Int[(c+d*Tan[e+f*x])^(n-2)*Simp[a*c^2+a*d^2*(n-1)-b*c*d*n-d*(a*c*(n-2)+b*d*n)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && GtQ[n,1]
```

2: $\int \frac{1}{(a+b\,\text{Tan}[e+f\,x])\,(c+d\,\text{Tan}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq0\;\wedge\;a^2+b^2==0\;\wedge\;c^2+d^2\neq0$

Derivation: Algebraic expansion

Basis: $\frac{1}{(a+b\,z)\,(c+d\,z)}==\frac{b}{(b\,c-a\,d)\,(a+b\,z)}-\frac{d}{(b\,c-a\,d)\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq0\;\wedge\;a^2+b^2==0\;\wedge\;c^2+d^2\neq0$**, then**

$$\int\frac{1}{(a+b\,\text{Tan}[e+f\,x])\,(c+d\,\text{Tan}[e+f\,x])}\,dx\;\longrightarrow\;\frac{b}{b\,c-a\,d}\int\frac{1}{a+b\,\text{Tan}[e+f\,x]}\,dx-\frac{d}{b\,c-a\,d}\int\frac{1}{c+d\,\text{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[1/((a_.+b_.*tan[e_.+f_.*x_])*(c_.+d_.*tan[e_.+f_.*x_])),x_Symbol] :=
  b/(b*c-a*d)*Int[1/(a+b*Tan[e+f*x]),x] - d/(b*c-a*d)*Int[1/(c+d*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

3: $\int \frac{(c+d\,\text{Tan}[e+f\,x])^n}{a+b\,\text{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\ \wedge\ a^2+b^2==0\ \wedge\ c^2+d^2\neq 0\ \wedge\ n\not>0$

Derivation: Symmetric tangent recurrence 2b with $A\to 1,\ B\to 0,\ m\to -1$

Rule: If $b\,c-a\,d\neq 0\ \wedge\ a^2+b^2==0\ \wedge\ c^2+d^2\neq 0\ \wedge\ n\not>0$, then

$$\int \frac{(c+d\,\text{Tan}[e+f\,x])^n}{a+b\,\text{Tan}[e+f\,x]}\,dx\ \longrightarrow$$

$$-\frac{a\,(c+d\,\text{Tan}[e+f\,x])^{n+1}}{2\,f\,(b\,c-a\,d)\,(a+b\,\text{Tan}[e+f\,x])}+\frac{1}{2\,a\,(b\,c-a\,d)}\int(c+d\,\text{Tan}[e+f\,x])^n\,(b\,c+a\,d\,(n-1)-b\,d\,n\,\text{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(c_.+d_.*tan[e_.+f_.*x_])^n_/(a_+b_.*tan[e_.+f_.*x_]),x_Symbol] :=
  -a*(c+d*Tan[e+f*x])^(n+1)/(2*f*(b*c-a*d)*(a+b*Tan[e+f*x])) +
  1/(2*a*(b*c -a*d))*Int[(c+d*Tan[e+f*x])^n*Simp[b*c+a*d*(n-1)-b*d*n*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && Not[GtQ[n,0]]
```

4. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>1$

1: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>1\,\wedge\,n<-1$

Derivation: Symmetric tangent recurrence 1a with $A\to a,\ B\to b,\ m\to m-1$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>1\,\wedge\,n<-1$**, then**

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx\,dx\;\longrightarrow$$

$$-\frac{a^2\,(b\,c-a\,d)\,(a+b\,\text{Tan}[e+f\,x])^{m-2}\,(c+d\,\text{Tan}[e+f\,x])^{n+1}}{d\,f\,(b\,c+a\,d)\,(n+1)}+\frac{a}{d\,(b\,c+a\,d)\,(n+1)}$$

$$\int (a+b\,\text{Tan}[e+f\,x])^{m-2}\,(c+d\,\text{Tan}[e+f\,x])^{n+1}\,\Big(b\,(b\,c\,(m-2)-a\,d\,(m-2\,n-4))+\big(a\,b\,c\,(m-2)+b^2\,d\,(n+1)-a^2\,d\,(m+n-1)\big)\,\text{Tan}[e+f\,x]\Big)\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  -a^2*(b*c-a*d)*(a+b*Tan[e+f*x])^(m-2)*(c+d*Tan[e+f*x])^(n+1)/(d*f*(b*c+a*d)*(n+1)) +
  a/(d*(b*c+a*d)*(n+1))*Int[(a+b*Tan[e+f*x])^(m-2)*(c+d*Tan[e+f*x])^(n+1)*
    Simp[b*(b*c*(m-2)-a*d*(m-2*n-4))+(a*b*c*(m-2)+b^2*d*(n+1)-a^2*d*(m+n-1))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && GtQ[m,1] && LtQ[n,-1] && (IntegerQ[m] || IntegersQ[
```

2. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0\,\land\,m>1\,\land\,n\not<-1$

1: $\int \frac{(a+b\,\text{Tan}[e+f\,x])^{3/2}}{c+d\,\text{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0$

Derivation: Algebraic expansion

- **Basis: If** $a^2+b^2==0\,\land\,c^2+d^2\neq 0$, **then** $\frac{(a+b\,z)^{3/2}}{c+d\,z}==\frac{2\,a^2\,\sqrt{a+b\,z}}{a\,c-b\,d}-\frac{\left(2\,b\,c\,d+a\,\left(c^2-d^2\right)\right)\,(a-b\,z)\,\sqrt{a+b\,z}}{a\,\left(c^2+d^2\right)\,(c+d\,z)}$

Note: If $a^2+b^2==0\,\land\,c^2+d^2\neq 0$, **then** $a\,c-b\,d\neq 0$.

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0$, **then**

$$\int \frac{(a+b\,\text{Tan}[e+f\,x])^{3/2}}{c+d\,\text{Tan}[e+f\,x]}\,dx\,\rightarrow\,\frac{2\,a^2}{a\,c-b\,d}\int\sqrt{a+b\,\text{Tan}[e+f\,x]}\,dx-\frac{2\,b\,c\,d+a\,\left(c^2-d^2\right)}{a\,\left(c^2+d^2\right)}\int\frac{(a-b\,\text{Tan}[e+f\,x])\,\sqrt{a+b\,\text{Tan}[e+f\,x]}}{c+d\,\text{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^(3/2)/(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  2*a^2/(a*c-b*d)*Int[Sqrt[a+b*Tan[e+f*x]],x] -
   (2*b*c*d+a*(c^2-d^2))/(a*(c^2+d^2))*Int[(a-b*Tan[e+f*x])*Sqrt[a+b*Tan[e+f*x]]/(c+d*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

2: $\int \frac{(a+b\,\text{Tan}[e+f\,x])^{3/2}}{\sqrt{c+d\,\text{Tan}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0$

Derivation: Algebraic expansion

- **Basis: If** $a^2+b^2==0$, **then** $(a+b\,z)^{3/2}==2\,a\,\sqrt{a+b\,z}+\frac{b}{a}\,(b+a\,z)\,\sqrt{a+b\,z}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0$, **then**

$$\int \frac{(a+b\,\text{Tan}[e+f\,x])^{3/2}}{\sqrt{c+d\,\text{Tan}[e+f\,x]}}\,dx\,\rightarrow\,2\,a\int\frac{\sqrt{a+b\,\text{Tan}[e+f\,x]}}{\sqrt{c+d\,\text{Tan}[e+f\,x]}}\,dx+\frac{b}{a}\int\frac{(b+a\,\text{Tan}[e+f\,x])\,\sqrt{a+b\,\text{Tan}[e+f\,x]}}{\sqrt{c+d\,\text{Tan}[e+f\,x]}}\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^(3/2)/Sqrt[c_.+d_.*tan[e_.+f_.*x_]],x_Symbol] :=
  2*a*Int[Sqrt[a+b*Tan[e+f*x]]/Sqrt[c+d*Tan[e+f*x]],x] +
  b/a*Int[(b+a*Tan[e+f*x])*Sqrt[a+b*Tan[e+f*x]]/Sqrt[c+d*Tan[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

3: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2=0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>1\,\wedge\,m+n-1\neq 0$

Derivation: Symmetric tangent recurrence 1b with $A\to a,\ B\to b,\ m\to m-1$

Note: This rule is applied when $m\in\mathbb{Z}$ **even if** n **is symbolic since the antiderivative can be expressed in terms of hypergeometric functions instead of requiring Appell functions.**

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2=0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>1\,\wedge\,m+n-1\neq 0$**, then**

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx\longrightarrow$$

$$\frac{b^2\,(a+b\,\text{Tan}[e+f\,x])^{m-2}\,(c+d\,\text{Tan}[e+f\,x])^{n+1}}{d\,f\,(m+n-1)}+$$

$$\frac{a}{d\,(m+n-1)}\int (a+b\,\text{Tan}[e+f\,x])^{m-2}\,(c+d\,\text{Tan}[e+f\,x])^n\,(b\,c\,(m-2)+a\,d\,(m+2\,n)+(a\,c\,(m-2)+b\,d\,(3\,m+2\,n-4))\,\text{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b^2*(a+b*Tan[e+f*x])^(m-2)*(c+d*Tan[e+f*x])^(n+1)/(d*f*(m+n-1)) +
  a/(d*(m+n-1))*Int[(a+b*Tan[e+f*x])^(m-2)*(c+d*Tan[e+f*x])^n*
    Simp[b*c*(m-2)+a*d*(m+2*n)+(a*c*(m-2)+b*d*(3*m+2*n-4))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && IntegerQ[2*m] && GtQ[m,1] && NeQ[m+n-1,0] &&
  (IntegerQ[m] || IntegersQ[2*m,2*n])
```

5. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2=0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<0$

1. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2=0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<0\,\wedge\,n>0$

1: $\int (a+b\,\text{Tan}[e+f\,x])^m\,\sqrt{c+d\,\text{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2=0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<0$

Derivation: Symmetric tangent recurrence 2a with $A\to 1,\ B\to 0,\ n\to \frac{1}{2}$

Derivation: Symmetric tangent recurrence 2b with $A\to 0,\ B\to 1,\ n\to -\frac{1}{2}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2=0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<0$**, then**

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,\sqrt{c+d\,\text{Tan}[e+f\,x]}\,dx\longrightarrow$$

$$-\frac{b\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,\sqrt{c+d\,\mathrm{Tan}[e+f\,x]}}{2\,a\,f\,m}+\frac{1}{4\,a^2\,m}\int\frac{(a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,(2\,a\,c\,m+b\,d+a\,d\,(2\,m+1)\,\mathrm{Tan}[e+f\,x])}{\sqrt{c+d\,\mathrm{Tan}[e+f\,x]}}\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*Sqrt[c_.+d_.*tan[e_.+f_.*x_]],x_Symbol] :=
  -b*(a+b*Tan[e+f*x])^m*Sqrt[c+d*Tan[e+f*x]]/(2*a*f*m) +
  1/(4*a^2*m)*Int[(a+b*Tan[e+f*x])^(m+1)*Simp[2*a*c*m+b*d+a*d*(2*m+1)*Tan[e+f*x],x]/Sqrt[c+d*Tan[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[m,0] && IntegersQ[2*m]
```

2: $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0\,\land\,m<0\,\land\,n>1$

Derivation: Symmetric tangent recurrence 2a with $A\to c,\ B\to d,\ n\to n-1$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0\,\land\,m<0\,\land\,n>1$**, then**

$$\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$

$$-\frac{(b\,c-a\,d)\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n-1}}{2\,a\,f\,m}+$$

$$\frac{1}{2\,a^2\,m}\int(a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n-2}\,(c\,(a\,c\,m+b\,d\,(n-1))-d\,(b\,c\,m+a\,d\,(n-1))-d\,(b\,d\,(m-n+1)-a\,c\,(m+n-1))\,\mathrm{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  -(b*c-a*d)*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n-1)/(2*a*f*m) +
  1/(2*a^2*m)*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^(n-2)*
    Simp[c*(a*c*m+b*d*(n-1))-d*(b*c*m+a*d*(n-1))-d*(b*d*(m-n+1)-a*c*(m+n-1))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[m,0] && GtQ[n,1] && (IntegerQ[m] || IntegersQ[2
```

2: $\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0\,\land\,m<0\,\land\,n\not>0$

Derivation: Symmetric tangent recurrence 2b with $A\to 1,\ B\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0\,\land\,m<0\,\land\,n\not>0$**, then**

$$\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$

$$\frac{a\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{2\,f\,m\,(b\,c-a\,d)}+$$

$$\frac{1}{2\,a\,m\,(b\,c-a\,d)}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n}\,(b\,c\,m-a\,d\,(2\,m+n+1)+b\,d\,(m+n+1)\,\mathrm{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(2*f*m*(b*c-a*d)) +
  1/(2*a*m*(b*c-a*d))*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n*
    Simp[b*c*m-a*d*(2*m+n+1)+b*d*(m+n+1)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[m,0] && (IntegerQ[m] || IntegersQ[2*m,2*n])
```

6: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,n>1\,\wedge\,m+n-1\neq 0$

Derivation: Symmetric tangent recurrence 3a with $A\to c,\ B\to d,\ n\to n-1$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,n>1\,\wedge\,m+n-1\neq 0$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$
$$\frac{d\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n-1}}{f\,(m+n-1)}-$$
$$\frac{1}{a\,(m+n-1)}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n-2}\cdot$$
$$\left(d\,(b\,c\,m+a\,d\,(-1+n))-a\,c^2\,(m+n-1)+d\,(b\,d\,m-a\,c\,(m+2\,n-2))\,\mathrm{Tan}[e+f\,x]\right)\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  d*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n-1)/(f*(m+n-1)) -
  1/(a*(m+n-1))*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n-2)*
    Simp[d*(b*c*m+a*d*(-1+n))-a*c^2*(m+n-1)+d*(b*d*m-a*c*(m+2*n-2))*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && GtQ[n,1] && NeQ[m+n-1,0] && (IntegerQ[n] || Integ
```

7: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,n<-1$

Derivation: Symmetric tangent recurrence 3b with $A\to 1,\ B\to 0$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0\,\wedge\,n<-1$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$

$$\frac{d\,(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{f\,(n+1)\,\left(c^2+d^2\right)}-$$

$$\frac{1}{a\,(n+1)\,\left(c^2+d^2\right)}\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}\,(b\,d\,m-a\,c\,(n+1)+a\,d\,(m+n+1)\,\mathrm{Tan}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  d*(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)/(f*(n+1)*(c^2+d^2)) -
  1/(a*(c^2+d^2)*(n+1))*Int[(a+b*Tan[e+f*x])^m*(c+d*Tan[e+f*x])^(n+1)*
    Simp[b*d*m-a*c*(n+1)+a*d*(m+n+1)*Tan[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[n,-1] && (IntegerQ[n] || IntegersQ[2*m,2*n])
```

8: $\int\frac{(a+b\,\mathrm{Tan}[e+f\,x])^m}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{(a+b\,z)^m}{c+d\,z}==\frac{a\,(a+b\,z)^m}{a\,c-b\,d}-\frac{d\,(a+b\,z)^m\,(b+a\,z)}{(a\,c-b\,d)\,(c+d\,z)}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0$**, then**

$$\int\frac{(a+b\,\mathrm{Tan}[e+f\,x])^m}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx\,\longrightarrow\,\frac{a}{a\,c-b\,d}\int(a+b\,\mathrm{Tan}[e+f\,x])^m\,dx-\frac{d}{a\,c-b\,d}\int\frac{(a+b\,\mathrm{Tan}[e+f\,x])^m\,(b+a\,\mathrm{Tan}[e+f\,x])}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_/(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  a/(a*c-b*d)*Int[(a+b*Tan[e+f*x])^m,x] -
  d/(a*c-b*d)*Int[(a+b*Tan[e+f*x])^m*(b+a*Tan[e+f*x])/(c+d*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

9: $\int\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,\sqrt{c+d\,\mathrm{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\sqrt{c+d\,z}==\frac{a\,c-b\,d}{a\,\sqrt{c+d\,z}}+\frac{d\,(b+a\,z)}{a\,\sqrt{c+d\,z}}$

Note: If $a^2+b^2==0\,\wedge\,c^2+d^2\neq 0$**, then** $a\,c-b\,d\neq 0$**.**

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2==0\,\wedge\,c^2+d^2\neq 0$**, then**

$$\int \sqrt{a+b\,\text{Tan}[e+f\,x]}\,\sqrt{c+d\,\text{Tan}[e+f\,x]}\,dx \to \frac{a\,c-b\,d}{a}\int \frac{\sqrt{a+b\,\text{Tan}[e+f\,x]}}{\sqrt{c+d\,\text{Tan}[e+f\,x]}}\,dx+\frac{d}{a}\int \frac{\sqrt{a+b\,\text{Tan}[e+f\,x]}\,(b+a\,\text{Tan}[e+f\,x])}{\sqrt{c+d\,\text{Tan}[e+f\,x]}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*tan[e_.+f_.*x_]]*Sqrt[c_.+d_.*tan[e_.+f_.*x_]],x_Symbol] :=
  (a*c-b*d)/a*Int[Sqrt[a+b*Tan[e+f*x]]/Sqrt[c+d*Tan[e+f*x]],x] +
  d/a*Int[Sqrt[a+b*Tan[e+f*x]]*(b+a*Tan[e+f*x])/Sqrt[c+d*Tan[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

10: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0$

Derivation: Integration by substitution

- **Basis: If** $a^2+b^2==0$, **then** $(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n==\frac{a\,b}{f}\,\text{Subst}\left[\frac{(a+x)^{m-1}\left(c+\frac{d\,x}{b}\right)^n}{b^2+a\,x},\,x,\,b\,\text{Tan}[e+f\,x]\right]\,\partial_x\,(b\,\text{Tan}[e+f\,x])$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2==0\,\land\,c^2+d^2\neq 0$, **then**

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx \to \frac{a\,b}{f}\,\text{Subst}\left[\int \frac{(a+x)^{m-1}\left(c+\frac{d\,x}{b}\right)^n}{b^2+a\,x}\,dx,\,x,\,b\,\text{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  a*b/f*Subst[Int[(a+x)^(m-1)*(c+d/b*x)^n/(b^2+a*x),x],x,b*Tan[e+f*x]] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

5. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0$

   1. $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,m>2$

      1: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,m>2\,\land\,n<-1$

Derivation: Tangent recurrence 1a with $A\to a^2,\ B\to 2\,a\,b,\ C\to b^2,\ m\to m-2$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,m>2\,\land\,n<-1$, **then**

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx \to$$

$$\frac{(b\,c-a\,d)^2\,(a+b\,\text{Tan}[e+f\,x])^{m-2}\,(c+d\,\text{Tan}[e+f\,x])^{n+1}}{d\,f\,(n+1)\,\left(c^2+d^2\right)}-$$

$$\frac{1}{d\,(n+1)\,(c^2+d^2)}\int (a+b\,\text{Tan}[e+f\,x])^{m-3}\,(c+d\,\text{Tan}[e+f\,x])^{n+1}\cdot$$
$$\left(a^2\,d\,(b\,d\,(m-2)-a\,c\,(n+1))+b\,(b\,c-2\,a\,d)\,(b\,c\,(m-2)+a\,d\,(n+1))-\right.$$
$$d\,(n+1)\,\left(3\,a^2\,b\,c-b^3\,c-a^3\,d+3\,a\,b^2\,d\right)\,\text{Tan}[e+f\,x]-$$
$$\left.b\,\left(a\,d\,(2\,b\,c-a\,d)\,(m+n-1)-b^2\,\left(c^2\,(m-2)-d^2\,(n+1)\right)\right)\,\text{Tan}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (b*c-a*d)^2*(a+b*Tan[e+f*x])^(m-2)*(c+d*Tan[e+f*x])^(n+1)/(d*f*(n+1)*(c^2+d^2)) -
  1/(d*(n+1)*(c^2+d^2))*Int[(a+b*Tan[e+f*x])^(m-3)*(c+d*Tan[e+f*x])^(n+1)*
    Simp[a^2*d*(b*d*(m-2)-a*c*(n+1))+b*(b*c-2*a*d)*(b*c*(m-2)+a*d*(n+1)) -
      d*(n+1)*(3*a^2*b*c-b^3*c-a^3*d+3*a*b^2*d)*Tan[e+f*x] -
      b*(a*d*(2*b*c-a*d)*(m+n-1)-b^2*(c^2*(m-2)-d^2*(n+1)))*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && GtQ[m,2] && LtQ[n,-1] && IntegerQ[2*m]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>2\,\wedge\,n\nless -1$

Derivation: Tangent recurrence 2a with $A\to a^2,\ B\to 2\,a\,b,\ C\to b^2,\ m\to m-2$

Note: This rule is applied when $m\in\mathbb{Z}$ even if n is symbolic since the antiderivative can be expressed in terms of hypergeometric functions instead of requiring Appell functions.

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>2\,\wedge\,n\nless -1\,\wedge\,(n\geq -1\,\vee\,m\in\mathbb{Z})$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$

$$\frac{b^2\,(a+b\,\mathrm{Tan}[e+f\,x])^{m-2}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{d\,f\,(m+n-1)}+$$

$$\frac{1}{d\,(m+n-1)}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m-3}\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,\cdot$$

$$\left(a^3\,d\,(m+n-1)-b^2\,(b\,c\,(m-2)+a\,d\,(1+n))+b\,d\,(m+n-1)\,\left(3\,a^2-b^2\right)\,\mathrm{Tan}[e+f\,x]-b^2\,(b\,c\,(m-2)-a\,d\,(3\,m+2\,n-4))\,\mathrm{Tan}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b^2*(a+b*Tan[e+f*x])^(m-2)*(c+d*Tan[e+f*x])^(n+1)/(d*f*(m+n-1)) +
  1/(d*(m+n-1))*Int[(a+b*Tan[e+f*x])^(m-3)*(c+d*Tan[e+f*x])^n*
     Simp[a^3*d*(m+n-1)-b^2*(b*c*(m-2)+a*d*(1+n))+b*d*(m+n-1)*(3*a^2-b^2)*Tan[e+f*x]-
       b^2*(b*c*(m-2)-a*d*(3*m+2*n-4))*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && IntegerQ[2*m] && GtQ[m,2] && (GeQ[n,-1] || Intege
  Not[IGtQ[n,2] && (Not[IntegerQ[m]] || EqQ[c,0] && NeQ[a,0])]
```

2. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<-1$

1. $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<-1\,\wedge\,0<n<2$

1: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<-1\,\wedge\,1<n<2$

Derivation: Tangent recurrence 1a with $A\to a,\ B\to b,\ C\to 0,\ m\to m-1$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<-1\,\wedge\,1<n<2$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$

$$\frac{(b\,c-a\,d)\,(a+b\,\text{Tan}[e+f\,x])^{m+1}\,(c+d\,\text{Tan}[e+f\,x])^{n-1}}{f\,(m+1)\,\left(a^2+b^2\right)}+$$
$$\frac{1}{(m+1)\,\left(a^2+b^2\right)}\int(a+b\,\text{Tan}[e+f\,x])^{m+1}\,(c+d\,\text{Tan}[e+f\,x])^{n-2}\,\cdot$$
$$\left(a\,c^2\,(m+1)+a\,d^2\,(n-1)+b\,c\,d\,(m-n+2)-\left(b\,c^2-2\,a\,c\,d-b\,d^2\right)\,(m+1)\,\text{Tan}[e+f\,x]-d\,(b\,c-a\,d)\,(m+n)\,\text{Tan}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  (b*c-a*d)*(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^(n-1)/(f*(m+1)*(a^2+b^2)) +
  1/((m+1)*(a^2+b^2))*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^(n-2)*
    Simp[a*c^2*(m+1)+a*d^2*(n-1)+b*c*d*(m-n+2)-(b*c^2-2*a*c*d-b*d^2)*(m+1)*Tan[e+f*x]-d*(b*c-a*d)*(m+n)*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[m,-1] && LtQ[1,n,2] && IntegerQ[2*m]
```

2: $\int(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq0\,\land\,a^2+b^2\neq0\,\land\,c^2+d^2\neq0\,\land\,m<-1\,\land\,n>0$

Derivation: Tangent recurrence 1a with $A\to1,\ B\to0,\ C\to0$

Derivation: Tangent recurrence 3b with $A\to a,\ B\to b,\ C\to0,\ m\to m-1$

Rule: If $b\,c-a\,d\neq0\,\land\,a^2+b^2\neq0\,\land\,c^2+d^2\neq0\,\land\,m<-1\,\land\,n>0$**, then**

$$\int(a+b\,\text{Tan}[e+f\,x])^m\,(c+d\,\text{Tan}[e+f\,x])^n\,dx\,\longrightarrow$$
$$\frac{b\,(a+b\,\text{Tan}[e+f\,x])^{m+1}\,(c+d\,\text{Tan}[e+f\,x])^{n}}{f\,(m+1)\,\left(a^2+b^2\right)}+$$
$$\frac{1}{(m+1)\,\left(a^2+b^2\right)}\int(a+b\,\text{Tan}[e+f\,x])^{m+1}\,(c+d\,\text{Tan}[e+f\,x])^{n-1}\,\cdot$$
$$\left(a\,c\,(m+1)-b\,d\,n-(b\,c-a\,d)\,(m+1)\,\text{Tan}[e+f\,x]-b\,d\,(m+n+1)\,\text{Tan}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n/(f*(m+1)*(a^2+b^2)) +
  1/((m+1)*(a^2+b^2))*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^(n-1)*
    Simp[a*c*(m+1)-b*d*n-(b*c-a*d)*(m+1)*Tan[e+f*x]-b*d*(m+n+1)*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && LtQ[m,-1] && GtQ[n,0] && IntegerQ[2*m]
```

2: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<-1\,\wedge\,(n<0\,\vee\,m\in\mathbb{Z})$

Derivation: Tangent recurrence 3a with $A\to 1,\ B\to 0,\ C\to 0$

Note: This rule is applied when $m\in\mathbb{Z}$ even if n is symbolic since the antiderivative can be expressed in terms of hypergeometric functions instead of requiring Appell functions.

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m<-1\,\wedge\,(n<0\,\vee\,m\in\mathbb{Z})$, then

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx\ \longrightarrow$$

$$\frac{b^2\,(a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n+1}}{f\,(m+1)\,\left(a^2+b^2\right)\,(b\,c-a\,d)}+$$

$$\frac{1}{(m+1)\,\left(a^2+b^2\right)\,(b\,c-a\,d)}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Tan}[e+f\,x])^n\cdot$$

$$\left(a\,(b\,c-a\,d)\,(m+1)-b^2\,d\,(m+n+2)-b\,(b\,c-a\,d)\,(m+1)\,\mathrm{Tan}[e+f\,x]-b^2\,d\,(m+n+2)\,\mathrm{Tan}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b^2*(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^(n+1)/(f*(m+1)*(a^2+b^2)*(b*c-a*d)) +
  1/((m+1)*(a^2+b^2)*(b*c-a*d))*Int[(a+b*Tan[e+f*x])^(m+1)*(c+d*Tan[e+f*x])^n*
    Simp[a*(b*c-a*d)*(m+1)-b^2*d*(m+n+2)-b*(b*c-a*d)*(m+1)*Tan[e+f*x]-b^2*d*(m+n+2)*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && IntegerQ[2*m] && LtQ[m,-1] && (LtQ[n,0] || Intege
  Not[ILtQ[n,-1] && (Not[IntegerQ[m]] || EqQ[c,0] && NeQ[a,0])]
```

3: $\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>1\,\wedge\,n>0$

Derivation: Tangent recurrence 2a with $A\to a\,c,\ B\to b\,c+a\,d,\ C\to b\,d,\ m\to m-1,\ n\to n-1$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0\,\wedge\,m>1\,\wedge\,n>0$**, then**

$$\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c+d\,\mathrm{Tan}[e+f\,x])^n\,dx\ \longrightarrow$$

$$\frac{b\,(a+b\,\mathrm{Tan}[e+f\,x])^{m-1}\,(c+d\,\mathrm{Tan}[e+f\,x])^n}{f\,(m+n-1)}+$$

$$\frac{1}{m+n-1}\int (a+b\,\mathrm{Tan}[e+f\,x])^{m-2}\,(c+d\,\mathrm{Tan}[e+f\,x])^{n-1}\,\cdot$$

$$\left(a^2\,c\,(m+n-1)-b\,(b\,c\,(m-1)+a\,d\,n)+\left(2\,a\,b\,c+a^2\,d-b^2\,d\right)(m+n-1)\,\mathrm{Tan}[e+f\,x]+b\,(b\,c\,n+a\,d\,(2\,m+n-2))\,\mathrm{Tan}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_.+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b*(a+b*Tan[e+f*x])^(m-1)*(c+d*Tan[e+f*x])^n/(f*(m+n-1)) +
  1/(m+n-1)*Int[(a+b*Tan[e+f*x])^(m-2)*(c+d*Tan[e+f*x])^(n-1)*
    Simp[a^2*c*(m+n-1)-b*(b*c*(m-1)+a*d*n)+(2*a*b*c+a^2*d-b^2*d)*(m+n-1)*Tan[e+f*x]+b*(b*c*n+a*d*(2*m+n-2))*Tan[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && GtQ[m,1] && GtQ[n,0] && IntegerQ[2*n]
```

4. $\int \frac{(a+b\,\text{Tan}[e+f\,x])^m}{c+d\,\text{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0$

1: $\int \frac{1}{(a+b\,\text{Tan}[e+f\,x])\,(c+d\,\text{Tan}[e+f\,x])}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{1}{(a+b\,z)\,(c+d\,z)}==\frac{a\,c-b\,d}{(a^2+b^2)\,(c^2+d^2)}+\frac{b^2\,(b-a\,z)}{(b\,c-a\,d)\,(a^2+b^2)\,(a+b\,z)}-\frac{d^2\,(d-c\,z)}{(b\,c-a\,d)\,(c^2+d^2)\,(c+d\,z)}$

- **Rule: If** $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0$**, then**

$$\int \frac{A+B\,\text{Tan}[e+f\,x]}{(a+b\,\text{Tan}[e+f\,x])\,(c+d\,\text{Tan}[e+f\,x])}\,dx\,\longrightarrow$$

$$\frac{(a\,c-b\,d)\,x}{(a^2+b^2)\,(c^2+d^2)}+\frac{b^2}{(b\,c-a\,d)\,(a^2+b^2)}\int\frac{b-a\,\text{Tan}[e+f\,x]}{a+b\,\text{Tan}[e+f\,x]}\,dx-\frac{d^2}{(b\,c-a\,d)\,(c^2+d^2)}\int\frac{d-c\,\text{Tan}[e+f\,x]}{c+d\,\text{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[1/((a_+b_.*tan[e_.+f_.*x_])*(c_.+d_.*tan[e_.+f_.*x_])),x_Symbol] :=
  (a*c-b*d)*x/((a^2+b^2)*(c^2+d^2)) +
  b^2/((b*c-a*d)*(a^2+b^2))*Int[(b-a*Tan[e+f*x])/(a+b*Tan[e+f*x]),x] -
  d^2/((b*c-a*d)*(c^2+d^2))*Int[(d-c*Tan[e+f*x])/(c+d*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

2: $\int \frac{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{\sqrt{a+b\,z}}{c+d\,z}==\frac{a\,c+b\,d+(b\,c-a\,d)\,z}{(c^2+d^2)\,\sqrt{a+b\,z}}-\frac{d\,(b\,c-a\,d)\,(1+z^2)}{(c^2+d^2)\,\sqrt{a+b\,z}\,(c+d\,z)}$

- **Rule: If** $b\,c-a\,d\neq 0\,\wedge\,a^2+b^2\neq 0\,\wedge\,c^2+d^2\neq 0$, **then**

$$\int \frac{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx\,\longrightarrow$$

$$\frac{1}{c^2+d^2}\int \frac{a\,c+b\,d+(b\,c-a\,d)\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}\,dx-\frac{d\,(b\,c-a\,d)}{c^2+d^2}\int \frac{1+\mathrm{Tan}[e+f\,x]^2}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,(c+d\,\mathrm{Tan}[e+f\,x])}\,dx$$

Program code:

```
Int[Sqrt[a_.+b_.*tan[e_.+f_.*x_]]/(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  1/(c^2+d^2)*Int[Simp[a*c+b*d+(b*c-a*d)*Tan[e+f*x],x]/Sqrt[a+b*Tan[e+f*x]],x] -
  d*(b*c-a*d)/(c^2+d^2)*Int[(1+Tan[e+f*x]^2)/(Sqrt[a+b*Tan[e+f*x]]*(c+d*Tan[e+f*x])),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

3: $\int \frac{(a+b\,\mathrm{Tan}[e+f\,x])^{3/2}}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0$

Derivation: Algebraic expansion

- **Basis:** $\frac{(a+b\,z)^{3/2}}{c+d\,z} = \frac{a^2\,c-b^2\,c+2\,a\,b\,d+(2\,a\,b\,c-a^2\,d+b^2\,d)\,z}{(c^2+d^2)\,\sqrt{a+b\,z}} + \frac{(b\,c-a\,d)^2\,(1+z^2)}{(c^2+d^2)\,\sqrt{a+b\,z}\,(c+d\,z)}$

- **Rule: If** $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0$**, then**

$$\int \frac{(a+b\,\mathrm{Tan}[e+f\,x])^{3/2}}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx \longrightarrow$$

$$\frac{1}{c^2+d^2}\int \frac{a^2\,c-b^2\,c+2\,a\,b\,d+\left(2\,a\,b\,c-a^2\,d+b^2\,d\right)\,\mathrm{Tan}[e+f\,x]}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}}\,dx+\frac{(b\,c-a\,d)^2}{c^2+d^2}\int \frac{1+\mathrm{Tan}[e+f\,x]^2}{\sqrt{a+b\,\mathrm{Tan}[e+f\,x]}\,\,(c+d\,\mathrm{Tan}[e+f\,x])}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^(3/2)/(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  1/(c^2+d^2)*Int[Simp[a^2*c-b^2*c+2*a*b*d+(2*a*b*c-a^2*d+b^2*d)*Tan[e+f*x],x]/Sqrt[a+b*Tan[e+f*x]],x] +
  (b*c-a*d)^2/(c^2+d^2)*Int[(1+Tan[e+f*x]^2)/(Sqrt[a+b*Tan[e+f*x]]*(c+d*Tan[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

4: $\int \frac{(a+b\,\mathrm{Tan}[e+f\,x])^{m}}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,m\notin\mathbb{Z}$

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{1}{c+d\,z} = \frac{c-d\,z}{c^2+d^2} + \frac{d^2\,(1+z^2)}{(c^2+d^2)\,(c+d\,z)}$

- **Rule: If** $b\,c-a\,d\neq 0\,\land\,a^2+b^2\neq 0\,\land\,c^2+d^2\neq 0\,\land\,m\notin\mathbb{Z}$**, then**

$$\int \frac{(a+b\,\mathrm{Tan}[e+f\,x])^{m}}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx \longrightarrow \frac{1}{c^2+d^2}\int (a+b\,\mathrm{Tan}[e+f\,x])^m\,(c-d\,\mathrm{Tan}[e+f\,x])\,dx+\frac{d^2}{c^2+d^2}\int \frac{(a+b\,\mathrm{Tan}[e+f\,x])^m\,\left(1+\mathrm{Tan}[e+f\,x]^2\right)}{c+d\,\mathrm{Tan}[e+f\,x]}\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_/(c_.+d_.*tan[e_.+f_.*x_]),x_Symbol] :=
  1/(c^2+d^2)*Int[(a+b*Tan[e+f*x])^m*(c-d*Tan[e+f*x]),x] +
  d^2/(c^2+d^2)*Int[(a+b*Tan[e+f*x])^m*(1+Tan[e+f*x]^2)/(c+d*Tan[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0] && Not[IntegerQ[m]]
```

5: $\int (a + b\,\text{Tan}[e + f\,x])^m\,(c + d\,\text{Tan}[e + f\,x])^n\,dx$ when $b\,c - a\,d \neq 0\ \land\ a^2 + b^2 \neq 0\ \land\ c^2 + d^2 \neq 0$

Derivation: Integration by substitution

Basis: $F[\text{Tan}[e + f\,x]] = \frac{1}{f}\,\text{Subst}\left[\frac{F[x]}{1+x^2},\ x,\ \text{Tan}[e + f\,x]\right]\,\partial_x\text{Tan}[e + f\,x]$

Rule: If $b\,c - a\,d \neq 0\ \land\ a^2 + b^2 \neq 0\ \land\ c^2 + d^2 \neq 0$, then

$$\int (a + b\,\text{Tan}[e + f\,x])^m\,(c + d\,\text{Tan}[e + f\,x])^n\,dx \longrightarrow \frac{1}{f}\,\text{Subst}\left[\int \frac{(a + b\,x)^m\,(c + d\,x)^n}{1 + x^2}\,dx,\ x,\ \text{Tan}[e + f\,x]\right]$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_*(c_+d_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/f*Subst[Int[(a+b*ff*x)^m*(c+d*ff*x)^n/(1+ff^2*x^2),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2+b^2,0] && NeQ[c^2+d^2,0]
```

Rules for integrands of the form $(a + b\,\text{Tan}[e + f\,x])^m\,(c\,(d\,\text{Tan}[e + f\,x])^p)^n$

1: $\int (a + b\,\text{Tan}[e + f\,x])^m\,(d\,\text{Cot}[e + f\,x])^n\,dx$ when $n \notin \mathbb{Z}\ \land\ m \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $m \in \mathbb{Z}$, then $(a + b\,\text{Tan}[z])^m = \frac{d^m\,(b+a\,\text{Cot}[z])^m}{(d\,\text{Cot}[z])^m}$

Rule: If $n \notin \mathbb{Z}\ \land\ m \in \mathbb{Z}$, then

$$\int (a + b\,\text{Tan}[e + f\,x])^m\,(d\,\text{Cot}[e + f\,x])^n\,dx \longrightarrow d^m \int (b + a\,\text{Cot}[e + f\,x])^m\,(d\,\text{Cot}[e + f\,x])^{n-m}\,dx$$

Program code:

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(d_./tan[e_.+f_.*x_])^n_,x_Symbol] :=
  d^m*Int[(b+a*Cot[e+f*x])^m*(d*Cot[e+f*x])^(n-m),x] /;
FreeQ[{a,b,d,e,f,n},x] && Not[IntegerQ[n]] && IntegerQ[m]
```

```
Int[(a_.+b_.*cot[e_.+f_.*x_])^m_.*(d_./cot[e_.+f_.*x_])^n_,x_Symbol] :=
  d^m*Int[(b+a*Tan[e+f*x])^m*(d*Tan[e+f*x])^(n-m),x] /;
FreeQ[{a,b,d,e,f,n},x] && Not[IntegerQ[n]] && IntegerQ[m]
```

2: $\int (a+b\,\text{Tan}[e+f\,x])^m\,(c\,(d\,\text{Tan}[e+f\,x])^p)^n\,dx$ when $n\notin\mathbb{Z}\land m\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{(c\,(d\,\text{Tan}[e+f\,x])^p)^n}{(d\,\text{Tan}[e+f\,x])^{n\,p}} == 0$
- **Rule: If** $n\notin\mathbb{Z}\land m\notin\mathbb{Z}$**, then**

$$\int (a+b\,\text{Tan}[e+f\,x])^m\,(c\,(d\,\text{Tan}[e+f\,x])^p)^n\,dx \longrightarrow \frac{c^{\text{IntPart}[n]}\,(c\,(d\,\text{Tan}[e+f\,x])^p)^{\text{FracPart}[n]}}{(d\,\text{Tan}[e+f\,x])^{p\,\text{FracPart}[n]}}\int (a+b\,\text{Tan}[e+f\,x])^m\,(d\,\text{Tan}[e+f\,x])^{n\,p}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*tan[e_.+f_.*x_])^m_.*(c_.*(d_.*tan[e_.+f_.*x_])^p_)^n_,x_Symbol] :=
  c^IntPart[n]*(c*(d*Tan[e + f*x])^p)^FracPart[n]/(d*Tan[e + f*x])^(p*FracPart[n])*
    Int[(a+b*Tan[e+f*x])^m*(d*Tan[e+f*x])^(n*p),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[n]] && Not[IntegerQ[m]]
```

```
Int[(a_.+b_.*cot[e_.+f_.*x_])^m_.*(c_.*(d_.*cot[e_.+f_.*x_])^p_)^n_,x_Symbol] :=
  c^IntPart[n]*(c*(d*Cot[e + f*x])^p)^FracPart[n]/(d*Cot[e + f*x])^(p*FracPart[n])*
    Int[(a+b*Cot[e+f*x])^m*(d*Cot[e+f*x])^(n*p),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[n]] && Not[IntegerQ[m]]
```